# Stabilized optimization via an NCL algorithm 

D. Ma, K. Judd,
D. Orban, M. Saunders

G-2017-108
December 2017

La collection Les Cahiers du GERAD est constituée des travaux de recherche menés par nos membres. La plupart de ces documents de travail a été soumis à des revues avec comité de révision. Lorsqu'un document est accepté et publié, le pdf original est retiré si c'est nécessaire et un lien vers l'article publié est ajouté.

Citation suggérée: Ma, Ding; Judd, Kenneth; Orban, Dominique; Saunders, Michael (Décembre 2017). Stabilized optimization via an NCL algorithm, Rapport technique, Les Cahiers du GERAD G-2017-108, GERAD, HEC Montréal, Canada.

Avant de citer ce rapport technique, veuillez visiter notre site Web (https://www.gerad.ca/fr/papers/G-2017-108) afin de mettre à jour vos données de référence, s'il a été publié dans une revue scientifique.

The series Les Cahiers du GERAD consists of working papers carried out by our members. Most of these pre-prints have been submitted to peer-reviewed journals. When accepted and published, if necessary, the original pdf is removed and a link to the published article is added.

Suggested citation: Ma, Ding; Judd, Kenneth; Orban, Dominique; Saunders, Michael (December 2017). Stabilized optimization via an NCL algorithm, Technical report, Les Cahiers du GERAD G-2017-108, GERAD, HEC Montréal, Canada.

Before citing this technical report, please visit our website (https:// www.gerad.ca/en/papers/G-2017-108) to update your reference data, if it has been published in a scientific journal.

La publication de ces rapports de recherche est rendue possible grâce au soutien de HEC Montréal, Polytechnique Montréal, Université McGill, Université du Québec à Montréal, ainsi que du Fonds de recherche du Québec - Nature et technologies.

Dépôt légal - Bibliothèque et Archives nationales du Québec, 2017 - Bibliothèque et Archives Canada, 2017

The publication of these research reports is made possible thanks to the support of HEC Montréal, Polytechnique Montréal, McGill University, Université du Québec à Montréal, as well as the Fonds de recherche du Québec - Nature et technologies.

Legal deposit - Bibliothèque et Archives nationales du Québec, 2017

- Library and Archives Canada, 2017

GERAD HEC Montréal
3000, chemin de la Côte-Sainte-Catherine Montréal (Québec) Canada H3T 2A7

Tél. : 514 340-6053
Téléc. : 514 340-5665
info@gerad.ca
www.gerad.ca

# Stabilized optimization via an NCL algorithm 

Ding Ma ${ }^{a}$

Kenneth Judd ${ }^{b}$
Dominique Orban ${ }^{c}$
Michael Saunders ${ }^{a}$
${ }^{a}$ Management Science and Engineering, Stanford University, Stanford, CA 94305-4026, USA
${ }^{b}$ Hoover Institution, Stanford University, Stanford, CA 94305-6010, USA
${ }^{c}$ GERAD \& Department of Mathematics and Industrial Engineering, École Polytechnique, Montréal (Québec) Canada, H3T 2A7
dingma, saunders@stanford.edu
judd@hoover.stanford.edu
dominique.orban@gerad.ca
saunders@stanford.edu

December 2017
Les Cahiers du GERAD
G-2017-108
Copyright © 2017 GERAD, Ma, Judd, Orban, Saunders

[^0]The authors are exclusively responsible for the content of their research papers published in the series Les Cahiers du GERAD. Copyright and moral rights for the publications are retained by the authors and the users must commit themselves to recognize and abide the legal requirements associated with these rights. Thus, users:

- May download and print one copy of any publication from the public portal for the purpose of private study or research;
- May not further distribute the material or use it for any profitmaking activity or commercial gain;
- May freely distribute the URL identifying the publication.

If you believe that this document breaches copyright please contact us providing details, and we will remove access to the work immediately and investigate your claim.

Abstract: For optimization problems involving many nonlinear inequality constraints, we extend the boundconstrained (BCL) and linearly-constrained (LCL) augmented-Lagrangian approaches of LANCELOT and MINOS to an algorithm that solves a sequence of nonlinearly constrained augmented Lagrangian subproblems whose nonlinear constraints satisfy the LICQ everywhere. The NCL algorithm is implemented in AMPL and tested on large instances of a tax policy model that could not be solved directly by any of the state-of-the-art solvers that we tested due to degeneracy. Algorithm NCL with IPOPT as subproblem solver proves to be effective, with IPOPT achieving warm starts on each subproblem.

Acknowledgments: We are extremely grateful to the developers of AMPL and IPOPT for making the development and evaluation of Algorithm NCL possible. We are especially grateful to Mehiddin Al-Baali and other organizers of the NAO-IV conference Numerical Analysis and Optimization at Sultan Qaboos University, Muscat, Oman, which brought the authors and AMPL developers together in January 2017.

## 1 Introduction

We consider constrained optimization problems of the form

| NCO | $\operatorname{minimize}$ <br> $x \in \mathbb{R}^{n}$ <br>  <br> subject to$\quad \phi(x) \geq 0, \quad A x \geq b, \quad \ell \leq x \leq u$, |
| :--- | :--- |

where $\phi(x)$ is a smooth nonlinear function, $c(x) \in \mathbb{R}^{m}$ is a vector of smooth nonlinear functions, and $A x \geq b$ is a placeholder for a set of linear inequality or equality constraints, with $x$ lying between lower and upper bounds $\ell$ and $u$.

In some applications where $m \gg n$, there may be more than $n$ constraints that are essentially active at a solution. The constraints do not satisfy the linear independence constraint qualification (LICQ), and general-purpose solvers are likely to have difficulty converging. Some form of regularization is required. We achieve this by adapting the augmented Lagrangian algorithm of the general-purpose optimization solver LANCELOT $[4,5,13]$ to derive a sequence of regularized subproblems denoted in the next section by $\mathrm{NC}_{k}$.

## 2 BCL, LCL, and NCL methods

The theory for the large-scale solver LANCELOT is best described in terms of the general optimization problem

$$
\begin{array}{|ll|}
\hline \text { NECB } & \operatorname{minimize}_{x \in \mathbb{R}^{n}} \quad \phi(x) \\
\text { subject to } c(x)=0, \quad \ell \leq x \leq u
\end{array}
$$

with nonlinear equality constraints and bounds. We let $x^{*}$ denote a local solution of NECB and $\left(y^{*}, z^{*}\right)$ denote associated multipliers. LANCELOT treats NECB by solving a sequence of bound-constrained subproblems of the form

$$
\begin{array}{|ll}
\mathrm{BC}_{k} \quad & \underset{x}{\operatorname{minimize}} L\left(x, y_{k}, \rho_{k}\right)=\phi(x)-y_{k}^{T} c(x)+\frac{1}{2} \rho_{k}\|c(x)\|^{2} \\
& \text { subject to } \ell \leq x \leq u,
\end{array}
$$

where $y_{k}$ is an estimate of the Lagrange multipliers $y^{*}$ for the equality constraints. This was called a boundconstrained Lagrangian (BCL) method by Friedlander and Saunders [8], in contrast to the LCL (linearly constrained Lagrangian) methods of Robinson [16] and MINOS [14], whose subproblems LC ${ }_{k}$ contain bounds as in $\mathrm{BC}_{k}$ and also linearizations of the equality constraints at the current point $x_{k}$ (including linear constraints).

In order to treat NCO with a sequence of $\mathrm{BC}_{k}$ subproblems, we convert the nonlinear inequality constraints to equalities to obtain

$$
\begin{array}{|ll}
\hline \mathrm{NCO}^{\prime} & \begin{array}{l}
\operatorname{minimize} \\
x, s \\
\text { subject to } \quad \\
\end{array} \quad c(x)-s=0, \quad A x \geq b, \quad \ell \leq x \leq u, \quad s \geq 0
\end{array}
$$

with corresponding subproblems (including linear constraints)

$$
\begin{array}{|ll}
\hline \mathrm{BC}_{k}^{\prime} \quad & \underset{x, s}{\operatorname{minimize}} L\left(x, y_{k}, \rho_{k}\right)=\phi(x)-y_{k}^{T}(c(x)-s)+\frac{1}{2} \rho_{k}\|c(x)-s\|^{2} \\
& \text { subject to } A x \geq b, \quad \ell \leq x \leq u, \quad s \geq 0 . \\
\hline
\end{array}
$$

We now introduce variables $r=-(c(x)-s)$ into $\mathrm{BC}_{k}^{\prime}$ to obtain the nonlinearly constrained Lagrangian (NCL) subproblem

| $\mathrm{NC}_{k}$ | $\operatorname{minimize}_{x, r}$ |
| :--- | :--- |
|  | $\phi(x)+y_{k}^{T} r+\frac{1}{2} \rho_{k}\\|r\\|^{2}$ |
|  | subject to |
|  | $c(x)+r \geq 0, \quad A x \geq b, \quad \ell \leq x \leq u$, |

in which $r$ serves to make the nonlinear constraints independent. Assuming existence of finite multipliers and feasibility, for $\rho_{k}>0$ and larger than a certain finite value, the NCL subproblems should cause $y_{k}$ to approach $y^{*}$ and most of the solution $\left(x_{k}^{*}, r_{k}^{*}, y_{k}^{*}, z_{k}^{*}\right)$ of $\mathrm{NC}_{k}$ to approach $\left(x^{*}, y^{*}, z^{*}\right)$, with $r_{k}^{*}$ approaching zero.

Problem $\mathrm{NC}_{k}$ is analogous to Friedlander and Orban's formulation for convex quadratic programs [7, Equation (3.2)]. See also Arreckx and Orban [2], where the motivation is the same as here, achieving reliability when the nonlinear constraints don't satisfy LICQ.

Note that for general problems NECB, the BCL and LCL subproblems contain linear constraints (bounds only, or linearized constraints and bounds). Our NCL formulation retains nonlinear constraints in the $\mathrm{NC}_{k}$ subproblems, but simplifies them by ensuring that they satisfy LICQ. On large problems, the additional variables $r \in \mathbb{R}^{m}$ in $\mathrm{NC}_{k}$ may be detrimental to active-set solvers like MINOS or SNOPT [9] because they increase the number of degrees of freedom (superbasic variables). Fortunately they are easily accommodated by interior methods, as our numerical results show for IPOPT [17, 10]. We trust that the same will be true for KNITRO [3, 12].

### 2.1 The BCL algorithm

The LANCELOT BCL method is summarized in Algorithm BCL. Each subproblem $\mathrm{BC}_{k}$ is solved with a specified optimality tolerance $\omega_{k}$, generating an iterate $x_{k}^{*}$ and the associated Lagrangian gradient $z_{k}^{*} \equiv \nabla L\left(x_{k}^{*}, y_{k}, \rho_{k}\right)$. If $\left\|c\left(x_{k}^{*}\right)\right\|$ is sufficiently small, the iteration is regarded as "successful" and an update to $y_{k}$ is computed from $x_{k}^{*}$. Otherwise, $y_{k}$ is not altered but $\rho_{k}$ is increased.

Key properties are that the subproblems are solved inexactly, the penalty parameter is increased only finitely often, and the multiplier estimates $y_{k}$ need not be assumed bounded. Under certain conditions, all iterations are eventually successful, the $\rho_{k}$ 's remain constant, the iterates converge superlinearly, and the algorithm terminates in a finite number of iterations.

```
Algorithm 1 BCL (Bound-Constrained Lagrangian Method for NECB)
    procedure \(\operatorname{BCL}\left(x_{0}, y_{0}, z_{0}\right)\)
        Set penalty parameter \(\rho_{1}>0\), scale factor \(\tau>1\), and constants \(\alpha, \beta>0\) with \(\alpha<1\).
        Set positive convergence tolerances \(\eta_{*}, \omega_{*} \ll 1\) and infeasibility tolerance \(\eta_{1}>\eta_{*}\).
        \(k \leftarrow 0\), converged \(\leftarrow\) false
        repeat
            \(k \leftarrow k+1\)
            Choose optimality tolerance \(\omega_{k}>0\) such that \(\lim _{k \rightarrow \infty} \omega_{k} \leq \omega_{*}\).
            Find \(\left(x_{k}^{*}, z_{k}^{*}\right)\) that solves \(\mathrm{BC}_{k}\) to within \(\omega_{k}\).
            if \(\left\|c\left(x_{k}^{*}\right)\right\| \leq \max \left(\eta_{*}, \eta_{k}\right)\) then
                \(y_{k}^{*} \leftarrow y_{k}-\rho_{k} c\left(x_{k}^{*}\right)\)
                \(x_{k} \leftarrow x_{k}^{*}, y_{k} \leftarrow y_{k}^{*}, z_{k} \leftarrow z_{k}^{*} \quad\) update solution estimates
                if \(\left(x_{k}, y_{k}, z_{k}\right)\) solves NECB to within \(\omega_{*}\), converged \(\leftarrow\) true
                \(\rho_{k+1} \leftarrow \rho_{k} \quad\) keep \(\rho_{k}\)
                \(\eta_{k+1} \leftarrow \eta_{k} /\left(1+\rho_{k+1}^{\beta}\right) \quad\) decrease \(\eta_{k}\)
            else
                \(\rho_{k+1} \leftarrow \tau \rho_{k} \quad\) increase \(\rho_{k}\)
                \(\eta_{k+1} \leftarrow \eta_{0} /\left(1+\rho_{k+1}^{\alpha}\right) \quad\) may increase or decrease \(\eta_{k}\)
            end if
        until converged
        \(x^{*} \leftarrow x_{k}, \quad y^{*} \leftarrow y_{k}, \quad z^{*} \leftarrow z_{k}\)
    end procedure
```

Note that at step 8 of Algorithm BCL, the inexact minimization would be typically carried out from the initial guess $\left(x_{k}^{*}, z_{k}^{*}\right)$. However, other initial points are possible. At step 12 , we say that $\left(x_{k}, y_{k}, z_{k}\right)$ solves NECB to within $\omega_{*}$ if the largest dual infeasibility is smaller than $\omega_{*}$.

```
Algorithm 2 NCL (Nonlinearly Constrained Lagrangian Method for NCO)
    procedure \(\operatorname{NCL}\left(x_{0}, r_{0}, y_{0}, z_{0}\right)\)
        Set penalty parameter \(\rho_{1}>0\), scale factor \(\tau>1\), and constants \(\alpha, \beta>0\) with \(\alpha<1\).
        Set positive convergence tolerances \(\eta_{*}, \omega_{*} \ll 1\) and infeasibility tolerance \(\eta_{1}>\eta_{*}\).
        \(k \leftarrow 0\), converged \(\leftarrow\) false
        repeat
            \(k \leftarrow k+1\)
            Choose optimality tolerance \(\omega_{k}>0\) such that \(\lim _{k \rightarrow \infty} \omega_{k} \leq \omega_{*}\).
            Find \(\left(x_{k}^{*}, r_{k}^{*}, y_{k}^{*}, z_{k}^{*}\right)\) that solves \(\mathrm{NC}_{k}\) to within \(\omega_{k}\).
            if \(\left\|r_{k}^{*}\right\| \leq \max \left(\eta_{*}, \eta_{k}\right)\) then
                    \(y_{k}^{*} \leftarrow y_{k}+\rho_{k} r_{k}^{*}\)
                \(x_{k} \leftarrow x_{k}^{*}, r_{k} \leftarrow r_{k}^{*}, y_{k} \leftarrow y_{k}^{*}, z_{k} \leftarrow z_{k}^{*} \quad\) update solution estimates
                    if \(\left(x_{k}, y_{k}, z_{k}\right)\) solves NCO to within \(\omega_{*}\), converged \(\leftarrow\) true
                    \(\rho_{k+1} \leftarrow \rho_{k} \quad\) keep \(\rho_{k}\)
                    \(\eta_{k+1} \leftarrow \eta_{k} /\left(1+\rho_{k+1}^{\beta}\right) \quad\) decrease \(\eta_{k}\)
                else
                \(\rho_{k+1} \leftarrow \tau \rho_{k} \quad\) increase \(\rho_{k}\)
                \(\eta_{k+1} \leftarrow \eta_{0} /\left(1+\rho_{k+1}^{\alpha}\right) \quad\) may increase or decrease \(\eta_{k}\)
                end if
            until converged
            \(x^{*} \leftarrow x_{k}, \quad r^{*} \leftarrow r_{k}, y^{*} \leftarrow y_{k}, \quad z^{*} \leftarrow z_{k}\)
    end procedure
```


### 2.2 The NCL algorithm

To derive a stabilized algorithm for problem NCO, we modify Algorithm BCL by introducing $r$ and replacing the subproblems $\mathrm{BC}_{k}$ by $\mathrm{NC}_{k}$. The resulting method is summarized in Algorithm NCL. The update to $y_{k}$ becomes $y_{k}^{*} \leftarrow y_{k}-\rho_{k}\left(c\left(x_{k}^{*}\right)-s_{k}^{*}\right)=y_{k}+\rho_{k} r_{k}^{*}$, the value satisfied by an optimal $y_{k}^{*}$ for subproblem $\mathrm{NC}_{k}$. Step 8 of Algorithm NCL would typically use $\left(x_{k}^{*}, r_{k}^{*}, y_{k}^{*}, z_{k}^{*}\right)$ as initial guess, and that is what we use in our implementation below.

## 3 An application: optimal tax policy

Some challenging test cases arise from the tax policy models described in [11]. With $x=(c, y)$, they take the form

| TAX | $\operatorname{maximize}_{c, y}$ | $\sum_{i} \lambda_{i} U^{i}\left(c_{i}, y_{i}\right)$ |
| :---: | :---: | :---: |
|  |  |  |
|  | subject to | $U^{i}\left(c_{i}, y_{i}\right)-U^{i}\left(c_{j}, y_{j}\right) \geq 0$ |
|  | $\lambda^{T}(y-c) \geq 0$ |  |
|  | $c, y \geq 0$, |  |
|  | for all $i, j$ |  |
|  |  |  |

where $c_{i}$ and $y_{i}$ are the consumption and income of taxpayer $i$, and $\lambda$ is a vector of positive weights. The utility functions $U^{i}\left(c_{i}, y_{i}\right)$ are each of the form

$$
U(c, y)=\frac{(c-\alpha)^{1-1 / \gamma}}{1-1 / \gamma}-\psi \frac{(y / w)^{1 / \eta+1}}{1 / \eta+1}
$$

where $w$ is the wage rate and $\alpha, \gamma, \psi$ and $\eta$ are taxpayer heterogeneities. More precisely, the utility functions are of the form

$$
U^{i, j, k, g, h}\left(c_{p, q, r, s, t}, y_{p, q, r, s, t}\right)=\frac{\left(c_{p, q, r, s, t}-\alpha_{k}\right)^{1-1 / \gamma_{h}}}{1-1 / \gamma_{h}}-\psi_{g} \frac{\left(y_{p, q, r, s, t} / w_{i}\right)^{1 / \eta_{j}+1}}{1 / \eta_{j}+1}
$$

where $(i, j, k, g, h)$ and ( $p, q, r, s, t$ ) run over $n a$ wage types, $n b$ elasticities of labor supply, $n c$ basic need types, $n d$ levels of distaste for work, and $n e$ elasticities of demand for consumption, with $n a, n b, n c, n d$, ne determining the size of the problem, namely $m=T(T-1)$ nonlinear constraints, $n=2 T$ variables, with $T:=n a \times n b \times n c \times n d \times n e$.

Table 1 summarizes results for a 4D example ( $n e=1$ and $\gamma_{1}=1$ ). The first term of $U(c, y)$ becomes $\log (c-\alpha)$, the limit as $\gamma \rightarrow 1$. Problem NCO and Algorithm NCL were formulated in the AMPL modeling language [6]. The solvers SNOPT [9] and IPOPT [17] were unable to solve NCO itself, but Algorithm NCL was successful with IPOPT solving the subproblems $\mathrm{NC}_{k}$. We use a default configuration of IPOPT with MUMPS [1] as symmetric indefinite solver to compute search directions. We set the optimality tolerance for IPOPT to $\omega_{k}=10^{-6}$ throughout, and specified warm starts for $k \geq 2$ using options warm_start_init_point=yes and mu_init $=1 \mathrm{e}-4$. These options greatly improved the performance of IPOPT on each subproblem compared to cold starts, for which mu_init=0.1. It is helpful that only the objective function of $\mathrm{NC}_{k}$ changes with $k$.

Table 1: NCL results on a 4D example with $n a, n b, n c, n d=11,3,3,2$, giving $m=39006, n=395$. Itns refers to IPOPT's primal-dual interior point method, and Time is seconds on an Apple iMac with 2.93 GHz Intel Core i7.

| $k$ | $\rho_{k}$ | $\eta_{k}$ | $\left\\|r_{k}^{*}\right\\|_{\infty}$ | $\phi\left(x_{k}^{*}\right)$ | Itns | Time |
| :---: | :---: | :---: | :---: | :---: | ---: | ---: |
| 1 | $10^{2}$ | $10^{-2}$ | $3.1 \mathrm{e}-03$ | $-2.1478532 \mathrm{e}+01$ | 125 | 42.8 |
| 2 | $10^{2}$ | $10^{-3}$ | $1.3 \mathrm{e}-03$ | $-2.1277587 \mathrm{e}+01$ | 18 | 6.5 |
| 3 | $10^{3}$ | $10^{-3}$ | $6.6 \mathrm{e}-04$ | $-2.1177152 \mathrm{e}+01$ | 27 | 9.1 |
| 4 | $10^{3}$ | $10^{-4}$ | $5.5 \mathrm{e}-04$ | $-2.1110210 \mathrm{e}+01$ | 31 | 10.8 |
| 5 | $10^{4}$ | $10^{-4}$ | $2.9 \mathrm{e}-04$ | $-2.1066664 \mathrm{e}+01$ | 57 | 24.3 |
| 6 | $10^{5}$ | $10^{-4}$ | $6.5 \mathrm{e}-05$ | $-2.1027152 \mathrm{e}+01$ | 75 | 26.8 |
| 7 | $10^{5}$ | $10^{-5}$ | $5.2 \mathrm{e}-05$ | $-2.1018896 \mathrm{e}+01$ | 130 | 60.9 |
| 8 | $10^{6}$ | $10^{-5}$ | $9.3 \mathrm{e}-06$ | $-2.1015295 \mathrm{e}+01$ | 159 | 81.8 |
| 9 | $10^{6}$ | $10^{-6}$ | $2.0 \mathrm{e}-06$ | $-2.1014808 \mathrm{e}+01$ | 139 | 70.0 |
| 10 | $10^{7}$ | $10^{-6}$ | $2.1 \mathrm{e}-07$ | $-2.1014800 \mathrm{e}+01$ | 177 | 97.6 |

For this example, problem NCO has $m=39006$ nonlinear inequality constraints and one linear constraint in $n=395$ variables $x=(c, y)$, and nonnegativity bounds. Subproblem $\mathrm{NC}_{k}$ has 39007 constraints and 39402 variables when $r$ is included. Fortunately $r$ does not affect the complexity of each IPOPT iteration, but greatly improves stability. In contrast, active-set methods like MINOS and SNOPT are very inefficient on the $\mathrm{NC}_{k}$ subproblems because the large number of inequality constraints leads to thousands of minor iterations, and the presence of $r$ (with no bounds) leads to thousands of superbasic variables. About $3.2 n$ constraints were within $10^{-6}$ of being active.

Table 2 summarizes results for a 5D example. The $\mathrm{NC}_{k}$ subproblems have $m=32220$ nonlinear constraints and $n=360$ variables, leading to 32581 variables including $r$. Again the options warm_start_init_point=yes and mu_init $=1 \mathrm{e}-4$ for $k \geq 2$ led to good performance by IPOPT on each subproblem. About $3 n$ constraints were within $10^{-6}$ of being active.

Table 2: NCL results on a 5D example with $n a, n b, n c, n d$, $n e=5,3,3,2,2$, giving $m=32220, n=360$.

| $k$ | $\rho_{k}$ | $\eta_{k}$ | $\left\\|r_{k}^{*}\right\\|_{\infty}$ | $\phi\left(x_{k}^{*}\right)$ | Itns | Time |
| :---: | :---: | :---: | :---: | :---: | ---: | ---: |
| 1 | $10^{2}$ | $10^{-2}$ | $7.0 \mathrm{e}-03$ | $-4.2038075 \mathrm{e}+02$ | 95 | 41.1 |
| 2 | $10^{2}$ | $10^{-3}$ | $4.1 \mathrm{e}-03$ | $-4.2002898 \mathrm{e}+02$ | 17 | 7.2 |
| 3 | $10^{3}$ | $10^{-3}$ | $1.3 \mathrm{e}-03$ | $-4.1986069 \mathrm{e}+02$ | 20 | 8.1 |
| 4 | $10^{4}$ | $10^{-3}$ | $4.4 \mathrm{e}-04$ | $-4.1972958 \mathrm{e}+02$ | 48 | 25.0 |
| 5 | $10^{4}$ | $10^{-4}$ | $2.2 \mathrm{e}-04$ | $-4.1968646 \mathrm{e}+02$ | 43 | 20.5 |
| 6 | $10^{5}$ | $10^{-4}$ | $9.8 \mathrm{e}-05$ | $-4.1967560 \mathrm{e}+02$ | 64 | 32.9 |
| 7 | $10^{5}$ | $10^{-5}$ | $6.6 \mathrm{e}-05$ | $-4.1967177 \mathrm{e}+02$ | 57 | 26.8 |
| 8 | $10^{6}$ | $10^{-5}$ | $4.2 \mathrm{e}-06$ | $-4.1967150 \mathrm{e}+02$ | 87 | 46.2 |
| 9 | $10^{6}$ | $10^{-6}$ | $9.4 \mathrm{e}-07$ | $-4.1967138 \mathrm{e}+02$ | 96 | 53.6 |

For much larger problems of this type, we found that it was helpful to reduce mu_init more often, as illustrated in Table 3. The $\mathrm{NC}_{k}$ subproblems here have $m=570780$ nonlinear constraints and $n=1512$ variables, leading to 572292 variables including $r$. Note that the number of NCL iterations is stable $(k \leq 10)$, and IPOPT performs well on each subproblem with decreasing mu_init. This time about $6.6 n$ constraints were within $10^{-6}$ of being active.

Note that the LANCELOT approach allows early subproblems to be solved less accurately. It may save time to set $\omega_{k}=\eta_{k}$ (say) rather than $\omega_{k}=\omega_{*}$ throughout.

Table 3: NCL results on a 5D example with $n a, n b, n c, n e, n e=21,3,3,2,2$, giving $m=570780, n=1512$.

| $k$ | $\rho_{k}$ | $\eta_{k}$ | $\left\\|r_{k}^{*}\right\\|_{\infty}$ | $\phi\left(x_{k}^{*}\right)$ | mu_init | Itns | Time |
| :---: | :---: | :---: | :---: | :---: | :---: | ---: | ---: |
| 1 | $10^{2}$ | $10^{-2}$ | $5.1 \mathrm{e}-03$ | $-1.7656816 \mathrm{e}+03$ | $10^{-1}$ | 825 | 7763.3 |
| 2 | $10^{2}$ | $10^{-3}$ | $2.4 \mathrm{e}-03$ | $-1.7648480 \mathrm{e}+03$ | $10^{-4}$ | 66 | 472.8 |
| 3 | $10^{3}$ | $10^{-3}$ | $1.3 \mathrm{e}-03$ | $-1.7644006 \mathrm{e}+03$ | $10^{-4}$ | 106 | 771.3 |
| 4 | $10^{4}$ | $10^{-3}$ | $3.8 \mathrm{e}-04$ | $-1.7639491 \mathrm{e}+03$ | $10^{-5}$ | 132 | 1347.0 |
| 5 | $10^{4}$ | $10^{-4}$ | $3.2 \mathrm{e}-04$ | $-1.7637742 \mathrm{e}+03$ | $10^{-5}$ | 229 | 2450.9 |
| 6 | $10^{5}$ | $10^{-4}$ | $8.6 \mathrm{e}-05$ | $-1.7636804 \mathrm{e}+03$ | $10^{-6}$ | 104 | 1096.9 |
| 7 | $10^{5}$ | $10^{-5}$ | $4.9 \mathrm{e}-05$ | $-1.7636469 \mathrm{e}+03$ | $10^{-6}$ | 143 | 1633.4 |
| 8 | $10^{6}$ | $10^{-5}$ | $1.5 \mathrm{e}-05$ | $-1.7636252 \mathrm{e}+03$ | $10^{-7}$ | 71 | 786.1 |
| 9 | $10^{7}$ | $10^{-5}$ | $2.8 \mathrm{e}-06$ | $-1.7636196 \mathrm{e}+03$ | $10^{-7}$ | 67 | 725.7 |
| 10 | $10^{7}$ | $10^{-6}$ | $5.1 \mathrm{e}-07$ | $-1.7636187 \mathrm{e}+03$ | $10^{-8}$ | 18 | 171.0 |

## 4 AMPL models, data, and scripts

Algorithm NCL has been implemented in the AMPL modeling language [6] and tested on problem TAX. The following sections list each relevant file. The files are available from [15].

### 4.1 Tax model

File pTax 5 Dncl .mod codes subproblem $\mathrm{NC}_{k}$ for problem TAX with five parameters $w, \eta, \alpha, \psi, \gamma$, using $\mu:=1 / \eta$. Note that for $U(c, y)$ in the objective and constraint functions, the first term $(c-\alpha)^{1-1 / \gamma} /(1-1 / \gamma)$ is replaced by a piecewise-smooth function that is defined for all values of $c$ and $\alpha$ (see [11]).

Primal regularization $\frac{1}{2} \delta\|(c, y)\|^{2}$ with $\delta=10^{-8}$ is added to the objective function to promote uniqueness of the minimizer. The vector $r$ is called R to avoid a clash with subscript r .

```
# pTax5Dncl.mod
# An NLP to solve a taxation problem with 5-dimensional types of tax payers.
#
# 29 Mar 2005: Original AMPL coding for 2-dimensional types by K. Judd and C.-L. Su.
# 20 Sep 2016: Revised by D. Ma and M. A. Saunders.
# 08 Nov 2016: 3D version created.
# 08 Dec 2016: 4D version created.
# 10 Mar 2017: Piece-wise smooth utility function created.
# 12 Nov 2017: pTax5Dncl.mod derived from pTax5D.mod.
# 08 Dec 2017: pTax5Dncl files added to multiscale website.
# Define parameters for agents (taxpayers)
param na; # number of types in wage
param nb; # number of types in eta
param nc; # number of types in alpha
param nd; # number of types in psi
param ne; # number of types in gamma
set A := 1..na; # set of wages
set B := 1..nb; # set of eta
set C := 1..nc; # set of alpha
set D := 1..nd; # set of psi
set E := 1..ne; # set of gamma
set T = {A,B,C,D,E}; # set of agents
# Define wages for agents (taxpayers)
param wmin; # minimum wage level
param wmax; # maximum wage level
param w {A}; # i, wage vector
param mu{B}; # j, mu = 1/eta# mu vector
param mu1{B}; # mu1[j] = mu[j] + 1
param alpha{C}; # k, ak vector for utility
param psi{D}; # g
param gamma{E}; # h
param lambda{A,B,C,D,E}; # distribution density
param epsilon;
param primreg default 1e-8; # Small primal regularization
```

```
var c{(i,j,k,g,h) in T} >= 0.1; # consumption for tax payer (i,j,k,g,h)
var y{(i,j,k,g,h) in T} >= 0.1; # income for tax payer (i,j,k,g,h)
var R{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
!(i=p and j=q and k=r and g=s and h=t)} >= -1e+20, <= 1e+20;
param kmax default 20;
param rhok
param rhofac
param rhomax
param etak
param etafac
param etamin
param rmax default 0;
param rmin default 0; # min r (for printing)
param rnorm default 0; # ||r||_inf
param rtol default 1e-6; # quit if biggest |r_i| <= rtol
param nT default 1; # nT = na*nb*nc*nd*ne
param m default 1; # nT*(nT-1) = no. of nonlinear constraints
param n default 1; # 2*nT = no. of nonlinear variables
param ck{(i,j,k,g,h) in T} default 0; # current variable c
param yk{(i,j,k,g,h) in T} default 0; # current variable y
param rk{(i,j,k,g,h) in T, (p,q,r,s,t) in T: # current variable r = - (c(x) - s)
!(i=p and j=q and k=r and g=s and h=t)} default 0;
param dk{(i,j,k,g,h) in T, (p,q,r,s,t) in T: # current dual variables (y_k)
!(i=p and j=q and k=r and g=s and h=t)} default 0;
minimize f:
sum{(i,j,k,g,h) in T}
(
(if c[i,j,k,g,h] - alpha[k] >= epsilon then
- lambda[i,j,k,g,h] *
((c[i,j,k,g,h] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])
- psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j])
else
- lambda[i,j,k,g,h] *
(- 0.5/gamma[h] * epsilon^(-1/gamma[h]-1) * (c[i,j,k,g,h] - alpha[k])^2
+ ( 1+1/gamma[h])* epsilon^(-1/gamma[h] ) * (c[i,j,k,g,h] - alpha[k])
+(1/(1-1/gamma[h]) - 1 - 0.5/gamma[h]) * epsilon^(1-1/gamma[h])
- psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j])
)
+ 0.5 * primreg * (c[i,j,k,g,h]^2 + y[i,j,k,g,h]^2)
)
+ sum{(i,j,k,g,h) in T, (p,q,r,s,t) in T: !(i=p and j=q and k=r and g=s and h=t)}
(dk[i,j,k,g,h,p,q,r,s,t] * R[i,j,k,g,h,p,q,r,s,t]
+ 0.5 * rhok * R[i,j,k,g,h,p,q,r,s,t]^2);
subject to
Incentive{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
!(i=p and j=q and k=r and g=s and h=t)}:
(if c[i,j,k,g,h] - alpha[k] >= epsilon then
(c[i,j,k,g,h] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])
- psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j]
else
- 0.5/gamma[h] *epsilon^(-1/gamma[h]-1)*(c[i,j,k,g,h] - alpha[k])^2
+(1+1/gamma[h])*epsilon^(-1/gamma[h] )*(c[i,j,k,g,h] - alpha[k])
+ (1/(1-1/gamma[h]) - 1 - 0.5/gamma[h])*epsilon^(1-1/gamma[h])
- psi[g]*(y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j]
)
- (if c[p,q,r,s,t] - alpha[k] >= epsilon then
(c[p,q,r,s,t] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])
- psi[g]*(y[p,q,r,s,t]/w[i])^mu1[j] / mu1[j]
else
- 0.5/gamma[h] *epsilon^(-1/gamma[h]-1)*(c[p,q,r,s,t] - alpha[k])^2
+(1+1/gamma[h])*epsilon^(-1/gamma[h] )*(c[p,q,r,s,t] - alpha[k])
+(1/(1-1/gamma[h]) - 1 - 0.5/gamma[h])*epsilon^(1-1/gamma[h])
- psi[g]*(y[p,q,r,s,t]/w[i])^mu1[j] / mu1[j]
)
```

```
+ R[i,j,k,g,h,p,q,r,s,t] >= 0;
Technology:
sum{(i,j,k,g,h) in T} lambda[i,j,k,g,h]*(y[i,j,k,g,h] - c[i,j,k,g,h]) >= 0;
```


### 4.2 Tax model data

File pTax5Dncl.dat provides data for a specific problem.

```
# pTax5Dncl.dat
# 08 Dec 2017: pTax5Dncl files added to multiscale website.
data;
let na := 5;
let nb := 3;
let nc := 3;
let nd := 2;
let ne := 2;
# Set up wage dimension intervals
let wmin := 2;
let wmax := 4;
let {i in A} w[i] := wmin + ((wmax-wmin)/(na-1))*(i-1);
data;
param mu :=
1 0.5
2 1
3 2 ;
# Define mu1
let {j in B} mu1[j] := mu[j] + 1;
data;
param alpha :=
1 0
2 1
3 1.5;
param psi :=
1 1
2 1.5;
param gamma :=
1 2
2 3;
# Set up 5 dimensional distribution
let {(i,j,k,g,h) in T} lambda[i,j,k,g,h] := 1;
# Choose a reasonable epsilon
let epsilon := 0.1;
```


### 4.3 Initial values

File pTax5Dinitial.run solves a simplified model to compute starting values for Algorithm NCL. The nonlinear inequality constraints are removed, and $y=c$ is enforced. This model solves easily with MINOS or SNOPT on all cases tried. Solution values are output to file p5Dinitial.dat.

```
# pTax5Dinitial.run
# 08 Dec 2017: pTax5Dncl files added to multiscale website.
# Define parameters for agents (taxpayers)
```

```
param na := 5; # number of types in wage
```

param na := 5; \# number of types in wage
param nb := 3; \# number of types in eta
param nb := 3; \# number of types in eta
param nc := 3; \# number of types in alpha
param nc := 3; \# number of types in alpha
param nd := 2; \# number of types in psi
param nd := 2; \# number of types in psi
param ne := 2; \# number of types in gamma
param ne := 2; \# number of types in gamma
set A := 1..na; \# set of wages
set A := 1..na; \# set of wages
set B := 1..nb; \# set of eta
set B := 1..nb; \# set of eta
set C := 1..nc; \# set of alpha
set C := 1..nc; \# set of alpha
set D := 1..nd; \# set of psi
set D := 1..nd; \# set of psi
set E := 1..ne; \# set of gamma
set E := 1..ne; \# set of gamma
set T = {A,B,C,D,E}; \# set of agents
set T = {A,B,C,D,E}; \# set of agents

# Define wages for agents (taxpayers)

# Define wages for agents (taxpayers)

param wmin := 2; \# minimum wage level
param wmin := 2; \# minimum wage level
param wmax := 4; \# maximum wage level
param wmax := 4; \# maximum wage level
param w {i in A} := wmin + ((wmax-wmin)/(na-1))*(i-1); \# wage vector
param w {i in A} := wmin + ((wmax-wmin)/(na-1))*(i-1); \# wage vector

# Choose a reasonable epsilon

# Choose a reasonable epsilon

param epsilon := 0.1;
param epsilon := 0.1;

# mu vector

# mu vector

param mu {B}; \# mu = 1/eta
param mu {B}; \# mu = 1/eta
param mui{B}; \# mui[j] = mu[j] + 1
param mui{B}; \# mui[j] = mu[j] + 1
param alpha {C};
param gamma {E};
param psi {D};
var c {(i,j,k,g,h) in T} >= 0.1;
var y {(i,j,k,g,h) in T} >= 0.1;
maximize f: sum{(i,j,k,g,h) in T}
if c[i,j,k,g,h] - alpha[k] >= epsilon then
(c[i,j,k,g,h] - alpha[k])^(1-1/gamma[h]) / (1-1/gamma[h])

- psi[g] * (y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j]
else
- 0.5/gamma[h] *epsilon^(-1/gamma[h]-1)*(c[i,j,k,g,h] - alpha[k])^2
+(1+1/gamma[h])*epsilon^(-1/gamma[h]) *(c[i,j,k,g,h] - alpha[k])
+(1/(1-1/gamma[h]) -1 - 0.5/gamma[h])*epsilon^(1-1/gamma[h])
- psi[g] * (y[i,j,k,g,h]/w[i])^mu1[j] / mu1[j];
subject to
Budget {(i,j,k,g,h) in T}: y[i,j,k,g,h] - c[i,j,k,g,h] = 0;
let {(i,j,k,g,h) in T} y[i,j,k,g,h] := i+1;
let {(i,j,k,g,h) in T} c[i,j,k,g,h] := i+1;
data;
param mu :=
1 0.5
2 1
3 2;


# Define mu1

let {j in B} mul[j] := mu[j] + 1;
data;
param alpha :=
1 0
2 1
3 1.5;
param psi :=
1 1
2 1.5;
param gamma :=
1 2
2 3;

```
```

option solver minos;
option solver snopt;
option show_stats 1;
option minos_options , \
summary_file=6 \
print_file=9 \
scale=no \
print_level=0 \
*minor_iterations=200 \
major_iterations=2000\
iterations=50000 \
optimality_tol=1e-7
*penalty=100.0
completion=full
*major_damp=0.1 \
superbasics_limit=3000\
solution=yes
*verify_level=3
';
option snopt_options , \
summary_file=6 \
print_file=9 \
scale=no \
print_level=0 \
major_iterations=2000\
iterations=50000 \
optimality_tol=1e-7 \
*penalty=100.0 \
superbasics_limit=3000\
solution=yes \
*verify_level=3 \
';
display na,nb,nc,nd,ne;
solve;
display na,nb,nc,nd,ne;
display y,c >p5Dinitial.dat;
close p5Dinitial.dat;

```

\subsection*{4.4 NCL implementation}

File pTax5Dnclipopt.run uses files
```

pTax5Dinitial.run
pTax5Dncl.mod
pTax5Dncl.dat
pTax5Dinitial.dat

```
to implement Algorithm NCL. Subproblems \(\mathrm{NC}_{k}\) are solved in a loop until \(\left\|r_{k}^{*}\right\|_{\infty} \leq \mathrm{rtol}=1 \mathrm{e}-6\), or \(\eta_{k}\) has been reduced to parameter etamin \(=1 \mathrm{e}-8\), or \(\rho_{k}\) has been increased to parameter rhomax \(=1 \mathrm{e}+8\). The loop variable \(k\) is called K to avoid a clash with subscript k in the model file.

Optimality tolerance \(\omega_{k}=10^{-6}\) is used throughout to ensure that the solution of the final subproblem \(\mathrm{NC}_{k}\) will be close to a solution of the original problem if \(\left\|r_{k}^{*}\right\|_{\infty}\) is small enough for the final \(k\) ( \(\left\|r_{k}^{*}\right\|_{\infty} \leq\) rtol \(=1 \mathrm{e}-6\) ).

IPOPT is used to solve each subproblem \(\mathrm{NC}_{k}\), with runtime options set to implement increasingly warm starts.

\footnotetext{
\# pTax5Dnclipopt.run
\# 08 Dec 2017: pTax5Dncl files added to multiscale website.
}
```

reset;
model pTax5Dinitial.run;
reset;
model pTax5Dncl.mod;
data pTax5Dncl.dat;
data; var include p5Dinitial.dat;
model;
option solver ipopt;
option show_stats 1;
option ipopt_options , \
dual_inf_tol=1e-6 \
max_iter=5000 \
';

# NCL method.

# kmax, rhok, rhofac, rhomax, etak, etafac, etamin, rtol

# are defined in the .mod file.

printf "NCLipopt log for pTax5D\n" > 5DNCLipopt.log;
display na, nb, nc, nd, ne, primreg > 5DNCLipopt.log;
printf " k rhok etak rnorm Obj\n" > 5DNCLipopt.log;
for {K in 1..kmax}
{ display na, nb, nc, nd, ne, primreg, K, kmax, rhok, etak;
if K == 2 then
{option ipopt_options \$ipopt_options
, warm_start_init_point=yes \
mu_init=1e-4 \
'};
if K == 4 then {option ipopt_options \$ipopt_options ' mu_init=1e-5'};
if K == 6 then {option ipopt_options \$ipopt_options , mu_init=1e-6'};
if K == 8 then {option ipopt_options \$ipopt_options , mu_init=1e-7'};
if K ==10 then {option ipopt_options \$ipopt_options , mu_init=1e-8'};
solve;
let rmax := max({(i,j,k,g,h) in T, (p,q,r,s,t) in T:
!(i=p and j=q and k=r and g=s and h=t)} R[i,j,k,g,h,p,q,r,s,t]);
let rmin := min({(i,j,k,g,h) in T, (p,q,r,s,t) in T:
!(i=p and j=q and k=r and g=s and h=t)} R[i,j,k,g,h,p,q,r,s,t]);
display na, nb, nc, nd, ne, primreg, K, rhok, etak, kmax;
display K, kmax, rmax, rmin;
let rnorm := max(abs(rmax), abs(rmin)); \# ||r||_inf
printf "%4i %9.1e %9.1e %9.1e %15.7e\n", K, rhok, etak, rnorm, f >> 5DNCLipopt.log;
close 5DNCLipopt.log;
if rnorm <= rtol then
{ printf "Stopping: rnorm is small\n"; display K, rnorm; break; }
if rnorm <= etak then \# update dual estimate dk; save new solution
{let {(i,j,k,g,h) in T, (p,q,r,s,t) in T:
!(i=p and j=q and k=r and g=s and h=t)}
dk[i,j,k,g,h,p,q,r,s,t] :=
dk[i,j,k,g,h,p,q,r,s,t] + rhok*R[i,j,k,g,h,p,q,r,s,t];
let {(i,j,k,g,h) in T} ck[i,j,k,g,h] := c[i,j,k,g,h];
let {(i,j,k,g,h) in T} yk[i,j,k,g,h] := y[i,j,k,g,h];
display K, etak;
if etak == etamin then { printf "Stopping: etak = etamin\n"; break; }
let etak := max(etak*etafac, etamin);
display etak;
}
else \# keep previous solution; increase rhok
{ let {(i,j,k,g,h) in T} c[i,j,k,g,h] := ck[i,j,k,g,h];
let {(i,j,k,g,h) in T} y[i,j,k,g,h] := yk[i,j,k,g,h];
display K, rhok;
if rhok == rhomax then { printf "Stopping: rhok = rhomax\n"; break; }
let rhok := min(rhok*rhofac, rhomax);

```
```

display rhok;
}
}
display c,y; display na, nb, nc, nd, ne, primreg, rhok, etak, rnorm;

# Count how many constraint are close to being active.

data;
let nT := na*nb*nc*nd*ne; let m := nT*(nT-1); let n := 2*nT;
let etak := 1.0001e-10;
printf "\n m = %8i\n n = %8i\n", m, n >> 5DNCLipopt.log;
printf "\n Constraints within tol of being active\n\n" >> 5DNCLipopt.log;
printf " tol count count/n\n" >> 5DNCLipopt.log;
for {K in 1..10}
{
let kmax := card{(i,j,k,g,h) in T, (p,q,r,s,t) in T:
!(i=p and j=q and k=r and g=s and h=t)
and Incentive[i,j,k,g,h,p,q,r,s,t].slack <= etak};
printf "%9.1e %8i %8.1f\n", etak, kmax, kmax/n >> 5DNCLipopt.log;
let etak := etak*10;
}
printf "Created 5DNCLipopt.log\n";

```

\section*{5 Conclusions}

This work has been illuminating in several ways as we sought to improve our ability to solve examples of problem TAX.
- Small examples of the tax model solve efficiently with MINOS and SNOPT, but eventually fail to converge as the problem size increases.
- IPOPT also solves small examples efficiently, but eventually starts requesting additional memory for the MUMPS sparse linear solver. The solver may freeze, or the iterations may diverge.
- The \(\mathrm{NC}_{k}\) subproblems are not suitable for MINOS or SNOPT because of the large number of variables \((x, r)\) and the resulting number of superbasic variables (although warm-starts are natural).
- It is often said that interior methods cannot be warm-started. Nevertheless, IPOPT has several runtime options that have proved to be extremely helpful for implementing Algorithm NCL. For the results obtained here, it has been sufficient to say that warm starts are wanted for \(k>1\), and that the IPOPT barrier parameter should be initialized at decreasing values for later \(k\) (where only the objective of subproblem \(\mathrm{NC}_{k}\) changes with \(k\) ).
- The numerical examples of Section 3 had \(3 n, 3 n\) and \(6.6 n\) constraints essentially active at the solution, yet were solved successfully. They suggest that the NCL approach with an interior method as subproblem solver can overcome LICQ difficulties on problems that could not be solved directly.

\section*{References}
[1] Patrick R. Amestoy, Iain S. Duff, Jean-Yves L'Excellent, and Jacko Koster. A fully asynchronous multifrontal solver using distributed dynamic scheduling. Equation SIAM Journal on Matrix Analysis and Applications, 23(1):15-41, 2001.
[2] S. Arreckx and D. Orban. A regularized factorization-free method for equality-constrained optimization. Technical Report GERAD G-2016-65, GERAD, Montréal, QC, Canada, 2016.
[3] Richard H. Byrd, Jorge Nocedal, and Richard A. Waltz. Knitro: An integrated package for nonlinear optimization. In G. Di Pillo and M. Roma, editors, Equation Large-Scale Nonlinear Optimization, pages 35-59. Springer US, Boston, MA, 2006.
[4] A. R. Conn, N. I. M. Gould, and Ph. L. Toint. A globally convergent augmented Lagrangian algorithm for optimization with general constraints and simple bounds. Equation SIAM J. Numer. Anal., 28:545-572, 1991.
[5] A. R. Conn, N. I. M. Gould, and Ph. L. Toint. Equation LANCELOT: A Fortran Package for Large-scale Nonlinear Optimization (Release A). Lecture Notes in Computation Mathematics 17. Springer Verlag, Berlin, Heidelberg, New York, London, Paris and Tokyo, 1992.
[6] R. Fourer, D. M. Gay, and B. W. Kernighan. Equation AMPL: A Modeling Language for Mathematical Programming. Brooks/Cole, Pacific Grove, second edition, 2002.
[7] M. P. Friedlander and D. Orban. A primal-dual regularized interior-point method for convex quadratic programs. Equation Math. Prog. Comp., 4(1):71-107, 2012.
[8] M. P. Friedlander and M. A. Saunders. A globally convergent linearly constrained Lagrangian method for nonlinear optimization. Equation SIAM J. Optim., 15(3):863-897, 2005.
[9] P. E. Gill, W. Murray, and M. A. Saunders. SNOPT: An SQP algorithm for large-scale constrained optimization. Equation SIAM Review, 47(1):99-131, 2005. SIGEST article.
[10] IPOPT open source NLP solver. https://projects.coin-or.org/Ipopt.
[11] K. L. Judd, D. Ma, M. A. Saunders, and C.-L. Su. Optimal income taxation with multidimensional taxpayer types. Working paper, Hoover Institution, Stanford University, 2017.
[12] KNITRO optimization software. https://www.artelys.com/tools/knitro_doc/2_userGuide.html.
[13] LANCELOT optimization software. http://www.numerical.rl.ac.uk/lancelot/blurb.html.
[14] B. A. Murtagh and M. A. Saunders. A projected Lagrangian algorithm and its implementation for sparse nonlinear constraints. Equation Math. Program. Study, 16:84-117, 1982.
[15] NCL. http://stanford.edu/group/SOL/multiscale/models/NCL/.
[16] S. M. Robinson. A quadratically-convergent algorithm for general nonlinear programming problems. Equation Math. Program., 3:145-156, 1972.
[17] A. Wächter and L. T. Biegler. On the implementation of a primal-dual interior point filter line search algorithm for large-scale nonlinear programming. Equation Math. Program., 106(1), 2006.```


[^0]:    Les textes publiés dans la série des rapports de recherche Les Cahiers du GERAD n'engagent que la responsabilité de leurs auteurs. Les auteurs conservent leur droit d'auteur et leurs droits moraux sur leurs publications et les utilisateurs s'engagent à reconnaître et respecter les exigences légales associées à ces droits. Ainsi, les utilisateurs:

    - Peuvent télécharger et imprimer une copie de toute publication du portail public aux fins d'étude ou de recherche privée;
    - Ne peuvent pas distribuer le matériel ou l'utiliser pour une activité à but lucratif ou pour un gain commercial;
    - Peuvent distribuer gratuitement l'URL identifiant la publication. Si vous pensez que ce document enfreint le droit d'auteur, contacteznous en fournissant des détails. Nous supprimerons immédiatement l'accès au travail et enquêterons sur votre demande.

